
Abstract

In this reportwe discussthespeci�cation,analysisandveri�cation of anautomatedpark-
ing garagein mCRL2,aprocessalgebrawith data.Weview theparkinggarageasasystem
thatweconceptuallydivide into threelayers:a logical layer, asafetylayerandahardware
abstractionlayer. This allows us to abstractfrom implementationdetails(hardwareab-
stractionlayer)andalgorithmdesign(logical layer). Instead,we areableto focuson the
speci�cationof a communicationinterfacebetweenthesetwo layersthatonly allows safe
systembehaviour. This interfaceconstitutesthesafetylayer. For thesafetylayer, we iden-
tify andformulatea numberof requirements.Theseareveri�ed andreportedon. We also
discusstheanalysisof thespeci�cationof thesafetylayerwith asimplecustomvisualiza-
tion tool. This tool wasimplementedasaplug-in to themCRL2toolsetandhelpedusgain
importantinsightsduringthespeci�cationandanalysisof thesafetylayer.

1 Intr oduction

During the past two decadesprocessalgebrashave becomeincreasinglypopular tools for
describingcomplex systemsthat interactwith their environment[5]. Suchsystemsdiffer
from classicbatchprocessesthat receive input, processthe input, produceoutputand then
terminate.Instead,theirbehaviour is continuouslyin�uencedby informationthatthey receive
from their environment.An automatedparkinggarageis a primeexampleof sucha system.
It is in operation24 hoursa day, 7 daysa week: it is continuouslyon standby, readyto stow
away andretrieve carsfor users.Furthermore,thestateof thegarageis constantlychanging
ascarsareaddedandremoved.At any point in time,thisstatein�uenceswhichoperationsare
possibleandhow they areexecuted.

Processalgebraallows for high-level descriptionsof systemsthat interactwith their en-
vironment.A systemis regardedasa numberof interactingprocessesthat togetherdescribe
its behaviour. In this way a mathematicalmodelis acquiredandthis is usedto provevarious
propertiesof the systemusingmathematicalproof techniques.The processalgebrausedin
this reportis mCRL2[3]. It succeedsandextends� CRL [4, 5].

With � CRL a largenumberof real-world systemshavebeenanalysedandveri�ed. Some
of thesuccessesthat � CRL hasbookedincludethedetectionof anunknown deadlockin the
mostcomplex variantof the sliding window protocol[1, 11], the identi�cation of a number
of errorsin a distributedlifting systemfor trucks[2] andthediscoveryof two errorsin a Java
distributedmemoryimplementation[9]. From all theseexperimentswe unfortunatelyhave

hadto formulatethe100%rule: in 100%of thecasesconsidered,systemsturn out to contain
moreor lessseriouserrors,whicharenot veryhardto detect.

An importantcharacteristicof mCRL2andits predecessor� CRL is theinclusionof data.
Experiencehasshown that in real-world systemsdatais of paramountimportance[6]. In
additionto theexecutionof actions,it is oftenthecasethatdatais storedandcommunicated,
thushaving a signi�cant in�uence onsystembehaviour.

In theremainderof thisreportwediscussthespeci�cationof anautomatedparkinggarage
in mCRL2. We alsodiscussthe analysisandveri�cation of the systemandthe techniques
used. In section2 we provide moredetailson the challengespresentedby this system. In
section3 we outlinetheapproachwe have takento addresstheseissuesandwe explain how
we conceptuallydivide the systeminto threelayers. This allows us to concentrateon one
layer, thesafetylayer, which we believe is essentialin verifying that thesystemis safe. We
follow thisby adiscussionof mCRL2andthemCRL2toolsetin section4. In sections5, 6 and
7 wespecifythehardwareabstractionlayer, thesafetylayerandthelogical layerrespectively.
We treatthehardwareabstractionlayerandthelogical layerterselybut revealmoredetailsof
thesafetylayer: we provide a speci�cation, formulatea numberof safetyrequirementsand
verify these.In section8 we describeasimplevisualizationplug-in for themCRL2toolsetas
well astheinsightswegainedfrom usingit. Finally, we draw conclusionsin section9.

2 Problemdescription

Theparkinggaragethatwe areconcernedwith wascommissionedby developersin Bremen,
Germany andwasdesignedby theDutchcompany CVSSAutomatedParking Systems. It will
be realisedbelow streetlevel in the basementof an existing building. Accessto the garage
will beprovidedwith a verticallift shaftthathasadoorwayat streetlevel. To usethefacility,
userswill drive their car throughthis doorway into thelift. After they have exited from their
carandthe lift, their carwill automaticallybe loweredto an intermediatelevel, rotated180�

horizontally, loweredto thebasementandstowedawayusinga numberof conveyor beltsand
shuttles.Whenauserwishesto retrieveacar, this samesystemof conveyor beltsandshuttles
will beusedto bring thecarto thelift from which it will bebroughtto streetlevel. Sincethe
carhadbeenrotatedbefore,it will now facethedirectionof thestreet.Theuserstepsinto the
caranddrivesaway throughthedoorway.

Thesystemwill provideanumberof securityandsafetychecksduringcheck-inandcheck-
out of a car. This includesreadinga transpondercardon thecarandcheckinga databaseof
registeredusersbeforeopeningthedoorway to the lift. As thecar is driveninto the lift, the
userwill beprovidedwith a numberof cuesto ensurethatthecaris positionedappropriately.
Therewill alsobea checkto ensurethat thehandbrake is engaged.Beforelowering thecar
to the basement,the lift will be scannedto ensurethat thereare no living beingspresent.
Whena userwishesto retrieve a car therewill be the necessarysecuritychecksto prevent
theft. Furthermore,therewill betraf�c lights outsidethedoorway to preventtraf�c jamsand
theinstallationwill beprovidedwith drivermotors,positionsensors,closedcircuit television
cameras,smokealarmsandsprinklers.

In subsequentdescriptionsweconsciouslyabstractfrom detailssuchasdrivermotors,po-
sitionsensorsandsoforth. Wealsorestrictourselvesto only theverticallift andthebasement
level parkinggarage.We do not take into accountthemechanismsput in placefor regulating
traf�c outsidethe lift, correctlypositioningthecar in the lift, or cuesto enterandleave the
lift. This is donein orderto tightly draw the boundsof our scopeandto focuson what we
believearetheessentialelementsthatfacilitatethesafebehaviour of thesystem.In doingso,
weavoid gettingboggeddown by implementationissues,hardwareor logistics.

Figure1: Floorplanof theparkinggarage,basementlevel

Fromourperspectivetheoperationof thesystemis initiatedevery timeacaris positioned
appropriatelyin the lift at streetlevel or whena requestfor a car is received. Physicallywe
view thesystemasconsistingof a singlelift anda parkinggarage.The lift canbe in oneof
threevertical positions: streetlevel, rotation level or basementlevel. At the rotation level,
the lift is ableto rotate180� horizontally. This is provided that thereareno carspositioned
immediatelyadjacentto the lift shaft(on eitherside)at basementlevel. The �oor of the lift
consistsof a conveyor belt. Whenthelift is at thebasementlevel this conveyor belt is ableto
movesideways(seethedescriptionbelow).

Themostcomplex andmostinterestingpartof thesystemis theparkinggarageat base-
mentlevel. Themovementof carsat this level is facilitatedby anumberof conveyorbeltsand
shuttles.This is illustratedin the�oorplan of thebasementin �gure 1.

As shown in the�gure, thegarageis dividedinto threerows(r1, r2 andr3)andtencolumns
(c1-c10).Conveyor beltsareportrayedby grey rectangleswith arrows on their endsandare
identi�ed by labelssuchasb r1a sh (below we elaboratefurther on the namingconvention
used).The arrows indicatetheir directionof movement.Columnsc1 andc10 containthree
shuttleseach. In eachof thesecolumnsoneshuttlemay be tilted on its long endfacingthe
wall. This resultsin an openpositionto which adjacentloweredshuttlesmaybe moved. A
tilted shuttlemayalsomove to a new row positionbehindloweredshuttles(this implies that
it is possiblefor two shuttlesto be in the samerow andcolumnpositionprovided that one
is tilted and the other lowered). In Figure1, black arrows indicatethe directionsin which
shuttlescanmove.Similar to thelift, everyshuttlealsocontainsaconveyorbelt thatcanmove
sideways.

The lift shaft is in row r1. Notice that it is not placedover a full position, but rather
intersectstwo columns(c6 andc7). This artefact is dueto the constructionof the building
in which the garageis to be installed(andhencebeyond the control of the engineerswho
designedthe garage).More importantly, this implies that it mustbe possibleto move cars
half-columndistancesin the �rst row. For this reasonevery columnin row r1 is alsodivided
into ana (left) andb (right) partasindicatedby thedashedlinesin �gure 1. We usethissame

conventionin thenamingof thebeltsin the�gure. Hence,b r1a shrefersto theconveyorbelt
of theshuttleon theleft-handsideof row r1, andsoforth.

It is possibleto move any subsetof conveyor beltsandany subsetof loweredshuttles.It
is alsopossibleto connectneighbouringconveyorbeltsto functionasasinglelargerconveyor
belt. For instance,conveyor belt b r3a shcanbeconnectedto b r3. Furthermore,oneshuttle
in c1 andoneshuttlein c10canbe tilted. A tilted shuttlecanbemovedindependentlyfrom
theloweredshuttlesin thatcolumn.

Thesystemhardwarecandeterminewhetherany (half-)positionis freeor occupied.For
any column in r1, it is possibleto determinethe statusof its a andb part. Note that this
implies that individual carson r1 cannotbe identi�ed by hardwaresensors.Sincewe only
wantto guaranteesafebehaviour, this doesnot impactour speci�cation. It is alsopossibleto
determinewhetherthereis a lowered,a tilted or no shuttleat all in any row of c1 andc10.
Furthermore,the currentheightof the lift canbe determinedandalsowhetherit is free or
occupied.This makesit possibleto geta snapshotof thesystemat any point in time during
operation.

Apart from not beinginvolvedin hardwaredesign,it is alsonot our goalto beconcerned
with algorithmdesign. Instead,our goal is to provide specialistsin algorithmdesignwith
aninterfaceto anabstractionof theunderlyinghardwarethatguaranteesthesafeandcorrect
operationthereof.Thisprovidesaclearseparationof concerns.Thealgorithmsneedto ensure
thatcarsareef�ciently stowedaway andretrieved.They mustalsoensurethatthegaragecan
be�lled to its maximumcapacityof 29 cars(this leavesonepositionfreeandallows for the
garageto operatein a fashionsimilar to a largeslidingpuzzle).Evenif thereareerrorsin such
algorithms,the interfaceshouldnot allow theparkinggarageor thecarsin it to getdamaged
duringtheirexecution.It needsto specifythenecessarychecksandrestrictionsthatguarantee
the executionof only safeor legal moves. The safetyinterfacemustalsobe ableto report
on thesuccessor failureof issuedcommands.We envision thatproperlydesignedalgorithms
will beableto respondto suchfeedbackin anappropriatefashion.

3 Conceptualsystemdesign

3.1 Ar chitecture

As alreadyalludedto above, our aim is to specifya safetylayer that sits betweeneventual
placementandretrieval algorithmsandtheabstracthardwareof theautomatedparkinggarage.
This layermustallow only safeor legal instructionsandreporton theirsuccessor failure.We
thereforeintroduceathree-layeredarchitectureconsistingof alogicallayer(LL), asafetylayer
(SL) andahardwareabstractionlayer(HAL) (see�gure 2). With thisconceptualdivisioninto
layers,thesafetylayerensuresthesafeoperationof thesystemindependentlyof theparticular
algorithmsthatareimplementedandwithout beingconcernedwith hardwareimplementation
issues.

3.2 Data

Thefollowing dataarecommunicatedbetweenthelayers:

� Event: this datatyperepresentseventsthatareoutsidethescopeof our design,but that
dohaveanimpacton thesystem.We identify thefollowing events:

– add car: a new carhasenteredthelift.

– remove car: acarhasbeenremovedfrom thelift.

Figure2: Conceptuallythesystemconsistsof threelayers

� InstructionSet: thisdatatyperepresentssetsof instructionsthatareto beexecutedcon-
currentlyby theHAL. It consistsof a numberof elementsof typeInstruction.

� Instruction: this datatype representsthe single instructionsthat the hardwareshould
execute.They shouldbeimplementedby theHAL (seesection5). Thereare5 different
instructions:

– move belts(bs:BeltSet,d: Direction,ms:MoveSize): this instructionspeci�esthat
thesetof beltsbsshouldbemovedin directiond with a distanceof sizems(half
or full).

– move shuttles(shs:ShuttleSet,o: ShuttleOrientation,d: Direction): this instruc-
tion speci�esthatthesetof shuttlesshsin orientationo (loweredor tilted) should
be moved in directiond with a distanceof onerow interval. The indicationof
orientationo is neededfor disambiguation,sinceit is possiblefor botha lowered
andtilted shuttleto bein thesameposition.

– tilt shuttle(p: ShuttlePosition, o: ShuttleOrientation): this instructionspeci�es
that the shuttlein positionp andorientationo shouldbe loweredor tilted to the
orientationthatis theoppositeof o.

– move lift(h: Height): this instructionspeci�esa new vertical positionh to move
thelift to.

– rotate lift : this instructionspeci�esthatthelift shouldberotatedby 180� horizon-
tally.

� Result: this datatype indicateswhetheran instructionhasbeenexecutedsuccessfully
(ok) or whetherit hasfailed(fail).

� GlobalState: this datatypere�ects thecurrentsystemstate.That is, for every position
whetherit is free or occupied(FloorState), for every shuttlewhetherit is loweredor
tilted (ShuttleState), andfor thelift its currentverticalpositionandwhetherit is freeor
occupied(LiftState).

A precisespeci�cationof all datatypescanbefoundin appendixB.

3.3 Actions

To facilitatecommunicationbetweenthedifferentlayersin our conceptualdesign,we intro-
ducethefollowing actions(alsosee�gure 2):

� occur(e: Event): this actionsigni�es the occurrenceof an event e. When the HAL
detectse, theSL is informedby theactionoccur(e). In turn, theSL informstheLL by
issuinganidenticaloccur(e)action.Notethatfor thesake of modularityof thedesign,
theLL is notdirectly informedby theHAL.

� req(is: InstructionSet): this actionallows the LL to requestthe executionof a setof
instructionsis. This requestis propagatedto the HAL via the SL. The notion of a
setof instructionsallows for the executionof multiple instructionsthat apply to non-
overlappingareasof the basement(for instance,it is possiblewith a singlerequestto
issuedifferentinstructionsfor moving conveyorbeltsaslongasthebeltsin questiondo
notoverlap).

� ack req(is: InstructionSet): whentheLL requeststheexecutionof a setof instructions
is, andthesearedeemedsafeby theSL, theSL issuesanack req(is)action.

� denyreq(is: InstructionSet): in thecasethattheSL deemsarequestfor asetof instruc-
tions is from the LL asunsafeit repliesby issuinga denyreq(is)action. Whenthis
happens,no furtherrequestsaremadeto theHAL by theSL.

� ack exec(is: InstructionSet,r: Result): uponcompletionof a setof instructionsis with
resultr, theHAL issuesanack exec(is,r) actionto theSL. In turn, theSL issuesthis
actionto theLL.

� req state: this actionis usedto requestthe currentglobal statefrom the HAL via the
SL.

� ack state(gs:GlobalState): this actionis usedto communicatethecurrentglobalstate
from theHAL to higherlayers.This is donein responseto a req stateaction.

A precisedescriptionof themostrelevantactionsandtheir usecanbefoundin appendixC.1.

4 mCRL2

In following sectionsweprovidemoredetailsof theactualspeci�cationof thesystemin terms
of the layeredarchitectureintroducedabove. Beforewe proceed,we introducethe mCRL2
speci�cationlanguageandtheaccompanying mCRL2toolset.

The mCRL2 languageis a successorto � CRL. The most important improvementson
� CRL is that the languageis compositional,that is large systemscanbe speci�ed in terms
of smallercomponents.It also hasa more advanceddataspeci�cation languagewhich is
higher-orderandprovidesconcretedatatypes.

Below we only introducethe languagefeaturesthat areusedfor the speci�cationof the
safetylayer. Thedataandprocesslanguagesaretreatedseparately. A rich text formatis used
asopposedto plaintext in theappendices.To overcomethedifferencesbetweentheseformats,
a translationtableis providedin appendixA. In theplain text formatfoundin theappendices,
the%-symbolindicatesthebeginningof acommentthatextendsto theendof theline.

4.1 Data language

The mCRL2 datalanguageis in a functional languagebasedon higher-order abstract data
types[7, 8]. Types,constructors,functionsandtheirde�nitions canbedeclared.For instance,
thefollowing declaresthetypeA with constructorsc;d, functionsf ; g andthede�nitions of
f ; g:

sort A;
cons c;d : A;
map f : A ! A ! A;

g : A ! A;
var x; y : A;
eqn f (c;x) = c;

f (d;x) = x;
g = f (c);

In theequations,variablescanbeusedandpatternmatching on theconstructorsof thedata
typeis catered.Notethatfunctiontypesare�rst-classcitizens:functionsmayreturnfunctions.

Typereferencescanbedeclared,for instancein

sort B = A;

B is a synonym for A. With this it is possibleto de�ne recursive types(seebelow).
Furthermore,where clausescanbeusedasanabbreviationmechanism,for example:

map h : A ! A ! A;
var x; y : A;
eqn h(x; y) = f (z; g(z)) whr z = g(x; y) end ;

As mentionedabove,mCRL2alsohasconcretedatatypes.Theseconsistof standard data
typesandoperationsaswell astypeconstructors. For theformer, wehave thefollowing:

� Booleans(B) with constantstrue, falseandoperators: , ^ , _, ) . For all typesthe
equalityoperator== , inequality6= andconditionalif areprovided. So for instance
the expressionc == c is equalto true, c == d to falseandif (true; c;d) to c. Also,
expressionsof typeB maybeusedasconditionsin equations,for instance:

var x; y : A;
eqn x == y ! f (x; y) = x

� Unboundedpositive,naturalandintegernumbers(P; N andZ) with relationaloperators
< , � , > , � , unarynegation� andbinaryarithmeticoperators+ , � , � , div , mo d.

Thereare two type constructors,of which the �rst one is a structured type. This is a
compactwayof de�ning a typetogetherwith constructorandprojectionfunctions.A sortMS
of machinestatescanbedeclaredby:

sort MS = struct o� j standbyj starting j running j broken;

Thesortof binarytreeswith numbersastheir leaveslookslike this:

sort T = struct leaf (value : N) j node(left : T; right : T );

This declarestypeT with constructorsleaf : N ! T andnode : T � T ! T andprojection
functionsvalue : T ! N andleft ; right : T ! T . Functionson T canbe de�ned using
patternmatchingon theconstructorsleaf andnode.

Finally, thereis a list typeconstructor. Thefollowing declaresa list containingelements
of typeA;

sort AL = List (A)

This list hasconstructors[ ] : AL and . : A � AL ! AL . Also operators/ , ++ , head,
tail , rhead andrtail areprovided togetherwith list enumeration[ ; : : : ; ]. The following
expressionsof typeAL areall equivalent:[ c;d;d ], c . [ d;d ], [ c;d ] / d and[ ] ++ [ c;d ]++ [ d ].

4.2 Processlanguage

Themostbasicnotionin themCRL2processlanguageis anaction.Actionsrepresentatomic
events.They canbedeclaredin thefollowing way:

act a; b;
c : B;
d : B � P;

Thisdeclaresactionsa; b;c andd. Here,a andbareparameterlessactions,c is anactionwith
a dataparameterof typeB, andd hastwo parametersof type B andP respectively. For the
abovedeclaration,a, c(1) andd(6; tr ue) arevalid actions.

Processexpressionsarecompositionsof actionsusinga numberof operators:

� Deadlockor inaction� , whichdoesnotdisplayany behaviour.

� Alternativecomposition,writtenasp+ q. Thisexpressionnon-deterministicallychooses
to executeprocessexpressionp or q.

� Sequentialcomposition,writtenp � q. Thisexpression�rst executesp followedby q.

� Conditional,written asb ! p;q, whereb is a dataexpressionof typeB. Theprocess
expressionbehavesasanif-then-elseconstruct.That is, if b is true thenp is executed,
elseq is executed.The elsepart is optional, that is b ! p is a valid expressionthat
behavesasb ! p; � .

� Summationover datatypes. The sumoperator
P

d:D F (d), with F a mappingfrom
datatype D with constructorsd0; : : : ; dn , n � 0, to processexpressions,behavesas
F (d0) + : : : + F (dn ).

� Processreferences,written as X (d1; : : : ; dn ), with n � 0. Theseare referencesto
variablesdeclaredby processequations,thatareintroducednext.

Figure3: Typical speci�cation,analysisandveri�cation cycleusingmCRL2

Using processexpressionswe canform processequations. Take for instancethe following
declaration,usingtheactiondeclarationabove:

proc X (x : B; y : P) = (a + b) � X (x; y)
+ c(x) � X (: x; p + 1)
+ p > 1 ! b� d(x; p) � X (false; p � 1);

This declaresprocessvariableX with two dataparametersof typeB andP. For dataterms
t : B andu : P, X (t; u) behavesastheright-handsideof theequationin which x andy are
replacedby t andu. ProcessvariableX is oftencalleda processandparameterst andu are
calledthestateof X .

A processspeci�cationneedsto beinitialised.For example:

init X (true; 1);

This initialisesprocessX with (true; 1) asits initial state.

4.3 mCRL2 toolset

Speci�cation,analysisandveri�cation of systemswith mCRL2is alsosupportedby atoolset.
Someof thesetools, andhow they �t into the speci�cation, analysisandveri�cation cycle
areillustratedin �gure 3. Oncethespeci�cationactivity starts,simulatortoolsallow theuser
to simulatepartsof the resultingstatespace.Essentiallythis tool allows the userto follow
labelledtransitionsfrom onesystemstateto another. At any particulartime the simulator
provides the userwith an overview of the currentsystemstateand also providesa listing
of all possibletransitionsfrom that state(see�gure 4). After a transitionis selected,the
tool updatesthesystemstateaccordingly, andprovidesan overview of thenew stateandall
transitionspossiblefrom it.

Onceusersaresatis�ed with the speci�cation, the instantiatortool generatesthe corre-
spondingstatespace.By thisstage,requirementsshouldhavebeenformalisedandarechecked
againston thestatespaceto determinewhetherany statesexist thatviolatethese.

5 Hardware abstraction layer (HAL)

WeassumethattheHAL servesasacoherentinterfaceto all individualhardwarecomponents
in the garage(driver motorsandso forth). The HAL receivesrequestsfor setsof instruc-
tions from theSL (actionreq). For eachsetof instructionstheHAL attemptsto executethe
individual instructionsandreportsbackon theresultof theattempt(actionack exec).

Figure4: Simulator

In orderfor thespeci�cationof theSL to yield the intendedresults,instructionsmustbe
implementedcorrectlyby theHAL. For instance,thefollowing mustoccurat hardwarelevel
whentheinstructionmove belts(bs,d, ms)is received:connectall theconveyorbeltsspeci�ed
in bs, engagetheappropriatedrivermotorsandmovethebeltsby adistancecorrespondingto
msin thedirectionindicatedby d. To ensurethat thedistancemovedcorrespondsto msthe
appropriatepositionsensorsneedto beutilised.

Apart from executinginstructions,theHAL alsoprovidestheSL with accessto a global
snapshotof thecurrentsystemstatebeforeandafter issuingany instructions.Sucha system
stateis constructedby theHAL usingsensorsthatmonitorthestateof every(half-)positionof
thebelts,everyshuttleandthelift.

6 Safetylayer (SL)

For thesafetylayerwe identify thesafetyrequirementsanddesigna speci�cationfor which
theseshouldbesatis�ed.After thatwe verify thatthis is indeedthecase.

6.1 Requirements

Thesafetyrequirementsof thesafetylayermaybesummarisedasfollows:

1. Conveyor belts:

(a) If a caris movedbetweenbelts,bothbeltsshouldmove in thesamedirection.

(b) Carsshouldnotbeableto moveinto walls.

(c) Carsshouldnotbeableto moveto abelt thatis notavailable(thatis ashuttlethat
is tilted, or thelift belt notbeingin its lowestposition).

2. Shuttles:

(a) Whenmoving shuttles,ashuttleshouldnotbeableto move into thewall.

(b) Whenmoving shuttles,othershuttlesmaynotbedamaged.

(c) Whenmoving shuttles,carsmaynotbedamaged.

(d) Whentilting a shuttle,carsmaynotbedamaged.

(e) Whentilting a shuttle,othershuttlesmaynotbedamaged.

3. Lift:

(a) Whenmoving thelift, carsmaynotbedamaged.

(b) Whenrotatingthelift, carsmaynotbedamaged.

6.2 Speci�cation

In orderto verify thattheSL allowsonly saferequeststo bepropagatedto theHAL, we have
speci�ed its behaviour in mCRL2.This speci�cationis providedin appendixB andC.1. The
following shouldbenoted:

� We did not take themessagepassingof thesystemstateor theeventsinto account,as
this doesnot impactsafety.

� We assumethatthereis at mostonesetof instructionsin thesystem.This is not a real
restriction,as it seemsnaturalthat communicationbetweenthe layersis muchfaster
thantheactualexecutionof instructions.

The informal requirementsintroducedabove arerelevant to the InstructionSetdatatype
(seesection5). In ourspeci�cationof theSLweuseanallowedfunctiononsetsof instructions
to determinewhethera setof instructionsis safe.A setof instructionsis is allowedif:

1. is speci�esat leastoneinstruction.

2. Theinstructionsin is do notoverlap, thatis theareason which theinstructionsoperate
arepairwisedisjoint suchthatit is safeto executetheinstructionsin is simultaneously.

3. Eachinstructionin is is allowed.

For the last condition,we also needan allowed function on individual instructions,which
formsthecoreof thede�nition.

Theinstructionmove belts(bs:BeltSet,d: Direction,ms:MoveSize)is allowedif:

1. bsspeci�esat leastoneconveyor belt.

2. All conveyor beltsin bsdirectly bordereachother(this alsoimplies that they mustbe
in thesamerow).

3. All conveyor beltsin bsareavailable(in particular, this appliesto beltson the lift and
onshuttles).

4. At leastonepositionof sizemsmustbefreeat theendof thesetof beltsspeci�ed,this
freepositionshouldbeon thesideindicatedby d.

5. In thecasethatthespeci�edbeltsarein row r1, theremustbenocarsuspendedhalf-way
betweenthetwo outerbeltsof bsandtheirneighbours,if any.

Theinstructionmove shuttles(shs:ShuttleSet,o: ShuttleOrientation,d: Direction)is allowed
if:

1. shsspeci�esat leastoneshuttle.

2. All shuttlesin shsbordereachother(this impliesthatthey areall in columnc1or c10).

3. All speci�ed shuttlesmust be available in the orientationspeci�ed by o (loweredor
tilted).

4. Thereis anopenpositionat theendof shsin orientationo anddirectiond, this ensures
thatthereis anopenpositionfor theshuttlesto moveto.

5. For every loweredr1 shuttles in shs, theremustbeno carthat is suspendedbetweens
andits borderingconveyor belt.

Theinstructiontilt shuttle(p:ShuttlePosition,o: ShuttleOrientation)is allowedif:

1. It is not thecasethatthereis botha loweredanda tilted shuttlein thepositionspeci�ed
by p.

2. If o is lowered,thereis nocaron theshuttle(fully or partially).

Theinstructionmove lift(h: Height)is allowedif:

1. Thetargetheightspeci�edby h is not thecurrentheight.

2. If the currentheightis basementlevel, theremustbe no carssuspendedhalf-way be-
tweenthelift andconveyor beltsoneithersideof thelift.

Theinstructionrotate lift is allowedif:

1. Thecurrentheightof thelift is rotationlevel.

2. Threehalf-positionsonbothsidesof thelift arefree(this to keepcarsin thesepositions
from beingdamagedby therotationmechanism).

A formalspeci�cationof theallowedfunctioncanbefoundin appendixB.

6.3 Simulation

Becauseof the enormousamountof possibleinstructionsetsthat canbe requestedandex-
ecuted,it is impossibleto do any simulation,let aloneveri�cation, on the speci�cation in
appendixC.1.For this reasonwe applythefollowing reduction:

Reduction1 Abstractfrom setsof instructionsby focusingonsingleinstructionsonly.

Thecorrespondingspeci�cationcanbefoundin appendixC.2. On theonehandthis abstrac-
tion is dangerous,becausesetsof instructionsareanessentialpartof thesystem.Ontheother
hand,thecoresafetyissueliesin theallowedfunctiononsingleinstructions.Furthermore,the
numberof possiblesystemcon�gurationsremainsthesame,sincetheresultof executingaset
of instructionsconcurrentlyis thesameasexecutingthemsequentially. This implies that in
thecorrespondingstatespacethenumberof statesremains�x ed,but thenumberof transitions
arereducedsubstantially.

Although the former reductionmakes it possibleto perform simulation, it is not very
effective. For this reason,we abstractfrom non-essentialmessages:

Reduction2 Abtract from requestsandacknowledgements.Instead,it is assumedthat in-
structionsareexecutedsuccessfullyby theHAL.

Thecorrespondingspeci�cationcanbefoundin appendixC.3.Whensimulating,onlyallowed
instructionscanbeexecuted.

Figure5: Reductionof the�oorplan

6.4 Veri�cation

We want to verify the requirementsby exploring the statespace.However, the statespace
correspondingto speci�cationafterapplyingthereductionaboveconsistsof 6; 4 � 1011 (640
billion) states,anda multiple of this in transitions. This is prohibitively large. Hence,we
applyonelastabstraction:

Reduction3 The numberof positionsof the beltsarereducedto the minimum that retains
thebehavioural characteristicsof theoriginal con�guration. This entailsthe following. The
positionson theconveyor beltsb r2 andb r3 arereducedto two positionseach(see�gure 5).
Also, beltsb r1a andb r1b arereducedto to 11

2 full positions(or 3 half positions)each.

Thecorrespondingspeci�cationcanbefoundin appendixD.1. Its statespacehas3; 3 � 106

(3; 3 million) statesand9; 8 � 107 (98million) transitions,which is manageable.
Theapproachwetake in verifying therequirementsdeviatesa little from thetypicalcycle

of �gure 3 on page9. Namely, we extendthe speci�cationwith error actionsthat areonly
enabledwhena requirementis violated. Hence,therequirementsareful�lled whenthestate
spacedoesnot containany erroractions.

The speci�cationextendedwith error actionscanbe found in appendixD.2. In this ap-
pendixa translationfrom thehigh-level requirementsof section6.1 to a lower level of detail
canalsobe found. Extracarehasbeentaken to specifytheenablingconditionsof theerror
actions. Namely the useof elementsof the de�nition of the allowed function needsto be
avoidedasmuchaspossible,sincemistakesin theoriginal speci�cationcouldcarryover to
theveri�cation.

Keepingthis observation in mind, we have extendedthe speci�cationandgeneratedthe
statespace. This statespacedoesnot containany error actions,which meansthat all the
requirementsareful�lled.

7 Logical layer (LL)

As mentionedbefore,theLL utilisestheSL by requestingthatsetsof instructionsbeexecuted.
In turn, the SL reportsbackto the LL in the form of ack req, denyreq or ack execactions.
It alsoinforms theLL of eventsthat have occurred(a new car to stow away or the removal
of an existing car). This allows for thedevelopmentof algorithmsby expertswho canplug
their speci�cationsinto theSL. Thesealgorithmsmaycontainerrorsthat trigger the request
of unsafeinstructions.Dueto thesafetylayer, suchinstructionsareharmless(but shouldbe
avoided),sincethey will beblocked.

Figure6: Visualizationplug-in

8 Visualization

During thespeci�cationweoftenresortedto simulatingthebehaviour of thesystemusingthe
mCRL2toolset(seesection4.3). Thesimulatortool is a greataid duringspeci�cationsince
it allows us to quickly andincrementallycheckwhetherour speci�cationdoesindeedresult
in the behaviour that we hadanticipated. This is opposedto generatingandexaminingan
entiretransitionsystemwhich is quitean expensive undertaking(dueto thephenomenonof
statespaceexplosionandeffort neededfor formulatingandcheckingformal requirements).
However, duringspeci�cationwe soonrealisedthat interpretingthe text-basedoutputof the
simulatoris timeconsuming,notentirelyintuitiveandproneto humanerror(especiallyasthe
amountof informationthatis storedin statesincreases).

In order to addressthe above problemand inspiredby othervisualizationinitiatives to
aid in systemanalysis[10, 12], we implementeda very simplevisualizationtool asa plug-in
to thesimulator. The visualizationtool getsthecurrentsystemstatefrom thesimulatorand
mapsthe informationonto a simple2D �oorplan of the parkinggarage(see�gure 6). This
visualizationusesvisualcuesto indicatetheverticallift positionandwhethera speci�c posi-
tion is occupied(red), free(green)or unavailable(gray). Tilted shuttlesarealsovisualised.
Insteadof interpretingthetext-basedrepresentationof thecurrentsystemstate,afterselecting
a new transition,thevisualizationis updatedandtheuseris ableto analysethesystemwith
this representation.

Webelievethatthismodeof analysissavedagreatamountof time. Dueto thefactthatwe
couldfollow carsasthey weretransporteddown thelift andmovedto new positionsusingthe
conveyor beltsandshuttleswe wereableto constructpotentiallydangerousscenarios.This
allowedusto identify andcorrectanumberof problemsearlyon. Theseincludedmistakeson
ourpartaswell asunknown complexitiesabouttheparkinggaragesetupin general.Although
all requirementscanbecheckedduringtheformalveri�cation stage,this restsontheassump-
tion thatall relevantquestionshavebeenidenti�ed andformalised.By visualizingthecurrent
stateit is alsopossibleto identify additionalissuesthatmaynot have beennoted.Moreover,
hadsoftwaresimplybeendesignedandimplementedfor thegaragewithout a detailedanaly-
sis,we believe thatsuchissuescouldhave easilycreptinto theimplementationof thegarage
despiteconsiderableprecautionon thepartof thedesigners.Unfortunately, weknow of a less
carefullydesignedgaragewherebothcarsandvital equipmenthavebeendamaged.

While usingour visualizationtool in designingthe systemwe discovereda numberof

Figure7: Mistakesidenti�ed with thevisualizationplug-in

problemsrelatedto the fact that carsmay be moved in half positionsin row r1. First, we
moveda cartowardthesidesof thegarageandpositionedit with onehalf on a conveyor belt
andonehalf on a shuttle. To our surprise,it waspossibleto subsequentlymove theshuttle,
literally tearingthe car in half (see�gure 7a)! This �rst bug we discoveredwasrelatively
easyto �x. A similar problemoccurredwhentwo carswerepositionedside-by-sideon the
lift, eachwith onehalf on a borderingconveyor belt. In this case,despiteour bestefforts of
explicitly checkingfor suchasituation,it waspossibleto movethelift up, tearingtwo carsin
half (see�gure 7b). This turnedout to bea harderproblemto solve.

We found the representationof our visualizationtool to be intuitively clearandeasyto
follow. In this casewere are fortunatethat thereexists suchan intuitive mappingto a vi-
sual representation.This is often not the case. For instance,for communicationprotocols,
thereis no“natural” visualrepresentationrequiringresearchersto comeupwith moreabstract
solutions.

9 Conclusions

In this reportwe have analysedanautomatedparkinggarage.We proposeda systemdesign
consistingof threelayers:a logical layer, a safetylayeranda hardwareabstractionlayer. We
havegivenaformalspeci�cationof thesafetylayerandafterapplyinganumberof reductions
we wereableto verify thesafetyrequirementsfor this speci�cation.This meansthatin every
valid con�gurationof thesystem,nodamagecanbedoneby executingunsafeinstructions.

We stronglyrecommendusingthespeci�cationof datatypesandprocessesdiscussedin
this reportasastartingpoint for theimplementationof softwareto operatetheparkinggarage
(appendicesB andC.1). This holdsespeciallyfor theconceptualthree-layerarchitectureand

thede�nitions of theallowedfunction.
We believe that our work comprisesthe essenceof the systemdesignregardingsafety.

However, thefollowing requiresfurtherinvestigationbeforestartingon theimplementation:

� In appendixC.1 a speci�cationof the systemis given in termsof setsof instructions.
Althoughwebelievethatthespeci�cationis correctfor thesesetsandalthoughwehave
simulatedonsomekey problemareasregardingsets,wehavenotbeenableto verify the
safety. Thismeansthatextracareshouldbetakenin theimplementationof theallowed
functiononsetsof instructionsin appendixB.

� We do not distinguishbetweenthe occurrenceof a recoverable andan unrecoverable
hardwarefailure.All failuresaretreatedasrecoverable.Furthermore,executionof aset
of instructionsonly givesoneresultwhich holdsfor all instructions.For moredetailed
errorhandling,theexecutionof individual instructionsshouldalsoreturnresults.That
is, after theexecutionof a setof instructions,someelementsmay returnok while the
othersmayreturnfail.

� In practise,the eventsadd car andremove car arenot atomic. They needto be split
up in a begin andanendpart. In orderto guaranteesafety, it shouldbe impossibleto
executea move lift instructionbetweenthebegin andendpartof anevent.

Althoughwehavenot investigatedtheef�ciency of theautomatedparkinggaragewefore-
seea performancechallengein termsof the timely stowing away and recovery of cars in
practise.However, we have taken careto ensurethat thedivision into threelayersdoesnot
degenerateperformance.In particular, thenotionof theconcurrentexecutionof instructions
addressesthis issue.

With regardto thegeneralsystemdesign,veri�cation andanalysiscycle(see�gure 3), we
want to stressthat simulationshouldnot be underestimated.All defectsin the speci�cation
werefoundusingsimulation.Also, simulationallowedusto identify andaddressinteresting
behavioural characteristicsthat would probablynot have beenincludedin the requirements
otherwise.In this way we wereableto gaininsight into thesystemin a way thatgoesfurther
thansimply listing andverifying requirements.Still, veri�cation remainsnecessary.

Finally, we wantto point out thedangerof usingthesamedatatypesfor bothsimulation
andveri�cation. This is dueto the possibility of errorsin the datatype speci�cation. The
approachwe took to circumvent this is to usedifferent functionsfor the veri�cation of the
mostcrucialfunctionsandsafetyrequirements.

Appendix A: Tableof mCRL2 symbols

In thespeci�cationof appendicesbelow, aplain text formatis usedasopposedto therich text
formatof section4. To make thetranslationfrom rich text symbolsto plain text symbols,the
following tableis provided:

Symbol Rich Plain
Booleans B Bool
positivenumbers P Pos
naturalnumbers N Nat
integers Z Int
arrow ! ->
inequality 6= !=
logical negation : !
conjunction ^ &&
disjunction _ ||
implication ) =>
smallerthanor equal � <=
greaterthanor equal � >=
list cons . |>
list snoc / <|
list concatenation ++ ++
sum

P
sum

Table1: Translationof mCRL2symbols

Appendix B: Data type speci®cation

All processspeci�cationsusethesamedatatypespeci�cation.For thisreason,weonly list the
generalpartonce.Thisspeci�cationis structuredasfollows. First all datatypesaredeclared.
After that,functionsfor eachdatatypearedeclaredandgroupedtogether.

sort
FloorRow = struct r1 | r2 | r3;
FloorCol = struct c1 | c2 | c3 | c4 | c5 | c6 | c7 | c8 | c9 | c10;
FloorPosPart = struct pa | pb;
FloorPos = struct pos_r1(FloorCol, FloorPosPart)

| pos_r2(FloorCol)
| pos_r3(FloorCol);

FloorPosList = List(FloorPos);
ShuttlePos = struct r1a | r2a | r3a | r1b | r2b | r3b;
ShuttlePosSet = List(ShuttlePos); % representing a set of ShuttlePos elements
ShuttleOrientation = struct lowered | tilted;
R1Belt = struct b_r1a_sh | b_r1a | b_r1lift | b_r1b | b_r1b_sh;
R2Belt = struct b_r2a_sh | b_r2 | b_r2b_sh;
R3Belt = struct b_r3a_sh | b_r3 | b_r3b_sh;
R1BeltSet = List(R1Belt); % representing a set of R1Belt elements
R2BeltSet = List(R2Belt); % representing a set of R2Belt elements
R3BeltSet = List(R3Belt); % representing a set of R3Belt elements
DirCol = struct col_inc | col_dec;
DirRow = struct row_inc | row_dec;
MoveSize = struct full | half;
LiftHeight = struct street | rotate | basement;
Area =

struct area(positions: FloorPosList, tilted_c1: Bool, tilted_c10: Bool);
Instruction = struct move_belts(R1BeltSet, DirCol, MoveSize)

| move_belts(R2BeltSet, DirCol, MoveSize)
| move_belts(R3BeltSet, DirCol, MoveSize)
| move_shuttles(ShuttlePosSet, ShuttleOrientation, DirRow)
| tilt_shuttle(ShuttlePos, ShuttleOrientation)
| move_lift(LiftHeight)
| rotate_lift;

InstructionSet = List(Instruction); % representing a set of instructions
Event = struct add_car | remove_car;
ExecResult = struct ok | fail;
OccState = struct free | occupied;
AvailState = struct avail | n_avail;
FloorState = FloorPos -> OccState;
ShuttleState = ShuttlePos # ShuttleOrientation -> AvailState;
LiftState = struct lsf_street | lso_street

| lsf_rotate | lso_rotate
| ls_basement;

GlobalState =
struct glob_state(fs: FloorState, shs: ShuttleState, ls: LiftState);

% FloorRow operations
map

index: FloorRow -> Pos;
% index(fra) is the index of fra
lt: FloorRow # FloorRow -> Bool;
% lt(fra,frb) indicates if fra is less than frb

var
fra, frb: FloorRow;

eqn
index(r1) = 1;
index(r2) = 2;
index(r3) = 3;
lt(fra, frb) = index(fra) < index(frb);

% FloorCol operations
map

index: FloorCol -> Pos;
% index(fca) is the index of fca
lt: FloorCol # FloorCol -> Bool;
% lt(fca,fcb) indicates if fca is less than fcb

var
fca, fcb: FloorCol;

eqn
index(c1) = 1;
index(c2) = 2;
index(c3) = 3;
index(c4) = 4;
index(c5) = 5;
index(c6) = 6;
index(c7) = 7;
index(c8) = 8;
index(c9) = 9;
index(c10) = 10;
lt(fca, fcb) = index(fca) < index(fcb);

% FloorPosPart operations
map

index: FloorPosPart -> Pos;
% index(ffpa) is the index of fppa
lt: FloorPosPart # FloorPosPart -> Bool;
% lt(fppa,fppb) indicates if fppa is less than fppb

var
fppa, fppb: FloorPosPart;

eqn
index(pa) = 1;
index(pb) = 2;
lt(fppa, fppb) = index(fppa) < index(fppb);

% FloorPos operations
map

row: FloorPos -> FloorRow;
% row(fpa) is the row of fpa
col: FloorPos -> FloorCol;
% col(fpa) is the column of fpa
part: FloorPos -> FloorPosPart;
% part(fpa) is the part of fpa
lt: FloorPos # FloorPos -> Bool;
% lt(fpa, fpb) indicates if fpa is less than fpb
lt_r: FloorPos # FloorPos -> Bool;
% lt_r(fpa, fpb) indicates if fpa is less than fpb, provided
% row(fpa) = row(fpb) (auxiliary function needed by lt for efficiency)
lt_rc: FloorPos # FloorPos -> Bool;
% lt_rc(fpa, fpb) indicates if fpa is less than fpb, provided
% row(fpa) = row(fpb) and col(fpa) = col(fpb)
% (auxiliary function needed by lt_r for efficiency)
gt: FloorPos # FloorPos -> Bool;
% gt(fpa, fpb) indicates if fpa is greater than fpb

var
fca: FloorCol;
fppa: FloorPosPart;
fpa, fpb: FloorPos;

eqn
row(pos_r1(fca, fppa)) = r1;
row(pos_r2(fca)) = r2;
row(pos_r3(fca)) = r3;
col(pos_r1(fca, fppa)) = fca;
col(pos_r2(fca)) = fca;
col(pos_r3(fca)) = fca;
part(pos_r1(fca, fppa)) = fppa;
row(fpa) == row(fpb) ->

lt(fpa, fpb) = lt_r(fpa, fpb);
row(fpa) != row(fpb) ->

lt(fpa, fpb) = lt(row(fpa), row(fpb));
col(fpa) == col(fpb) ->

lt_r(fpa, fpb) = lt_rc(fpa, fpb);
col(fpa) != col(fpb) ->

lt_r(fpa, fpb) = lt(col(fpa), col(fpb));
lt_rc(fpa, fpb) = row(fpa) == r1 && lt(part(fpa), part(fpb));
gt(fpa,fpb) = lt(fpb, fpa);

% FloorPosList operations
map

contains: FloorPos # FloorPosList -> Bool;
% contains(fpa, fps) indicates if fps contains fpa
overlap: FloorPosList # FloorPosList -> Bool;
% overlap(fps, fpt) indicates if the elements of fps and fpt overlap

rev: FloorPosList -> FloorPosList;
% rev(fps) is the reverse of list fps
grev: FloorPosList # FloorPosList -> FloorPosList;
% grev(fps, fpt) is the reverse of fps concatenated with fpt
% (auxiliary function needed for efficiency by rev)

var
fpa, fpb: FloorPos;
fps, fpt: FloorPosList;

eqn
contains(fpa, []) = false;
contains(fpa, fpa |> fps) = true;
fpa != fpb -> contains(fpa, fpb |> fps) = contains(fpa, fps);
overlap([], fpt) = false;
overlap(fpa |> fps, fpt) = contains(fpa, fpt) || overlap(fps, fpt);
rev(fps) = grev(fps, []);
grev([], fpt) = fpt;
grev(fpa |> fps, fpt) = grev(fps, fpa |> fpt);

% ShuttlePos operations
map

index: ShuttlePos -> Pos;
% index(spa) is the index of spa
row: ShuttlePos -> FloorRow;
% row(spa) is the row of spa
col: ShuttlePos -> FloorCol;
% col(spa) is the column of spa
lt: ShuttlePos # ShuttlePos -> Bool;
% lt(spa, spb) indicates if spa is less than spb
gt: ShuttlePos # ShuttlePos -> Bool;
% gt(spa, spb) indicates if spa is greater than spb
positions: ShuttlePos -> FloorPosList;
% positions(spa) represents the floor positions corresponding to spa in
% lowered position
has_neighbour: ShuttlePos # DirRow -> Bool;
% has_neighbour(spa, dr) indicates if spa has a successor in direction dr
neighbour: ShuttlePos # DirRow -> ShuttlePos;
% neighbour(spa, dr) is the neighbour of spa in directrion dr, provided it
% exists
connected: ShuttlePos # ShuttlePos -> Bool;
% connected(spa, spb) indicates if the bottom of spa is connected to the top
% of spb

var
spa, spb: ShuttlePos;

eqn
index(r1a) = 1;
index(r2a) = 2;
index(r3a) = 3;
index(r1b) = 4;
index(r2b) = 5;
index(r3b) = 6;
row(r1a) = r1;
row(r2a) = r2;
row(r3a) = r3;
row(r1b) = r1;
row(r2b) = r2;
row(r3b) = r3;
col(r1a) = c1;
col(r2a) = c1;
col(r3a) = c1;
col(r1b) = c10;
col(r2b) = c10;
col(r3b) = c10;
lt(spa, spb) = index(spa) < index(spb);
gt(spa, spb) = index(spa) > index(spb);
positions(r1a) = positions(b_r1a_sh);
positions(r2a) = positions(b_r2a_sh);
positions(r3a) = positions(b_r3a_sh);
positions(r1b) = positions(b_r1b_sh);
positions(r2b) = positions(b_r2b_sh);
positions(r3b) = positions(b_r3b_sh);
has_neighbour(spa, row_inc) = row(spa) != r3;
has_neighbour(spa, row_dec) = row(spa) != r1;
neighbour(r1a, row_inc) = r2a;
neighbour(r2a, row_inc) = r3a;

neighbour(r1b, row_inc) = r2b;
neighbour(r2b, row_inc) = r3b;
neighbour(r2a, row_dec) = r1a;
neighbour(r3a, row_dec) = r2a;
neighbour(r2b, row_dec) = r1b;
neighbour(r3b, row_dec) = r2b;
connected(spa, spb) =

succ(index(spa)) == index(spb) &&
col(spa) == col(spb);

% ShuttlePosSet operations
map

is_set_sps: ShuttlePosSet -> Bool;
% is_set_sps(sps) indicates if sps is a set
% In the remainder of the data equation section, for every ShuttlePosSet sps
% is_set_sps(sps) is a precondition.
contains: ShuttlePos # ShuttlePosSet -> Bool;
% contains(spa, sps) indicates if sps contains spa
shuttles: FloorCol -> ShuttlePosSet;
% shuttles(fc) represents the shuttle positions corresponding to fc
positions_sps: ShuttlePosSet -> FloorPosList;
% positions_sps(sps) represents the floor positions corresponding to sps
connected_sps: ShuttlePosSet -> Bool;
% connected_sps(sps) indicates if all elements of sps are connected
has_neighbour: ShuttlePosSet # DirRow -> Bool;
% has_neighbour(sps, dr) indicates is sps has a neighbour in direction dr
neighbour_sps: ShuttlePosSet # DirRow -> ShuttlePos;
% neighbour_sps(sps, dr) is the neighbour of sps in direction dr, provided it
% exists
add_neighbour: ShuttlePosSet # DirRow -> ShuttlePosSet;
% add_neighbour(sps, dr) represents the sps extended with its neighbour in
% direction dr, provided it exists

var
spa, spb: ShuttlePos;
sps: ShuttlePosSet;
fc: FloorCol;

eqn
is_set_sps([]) = true;
is_set_sps(spa |> []) = true;
is_set_sps(spa |> spb |> sps) =

lt(spa, spb) && is_set_sps(spb |> sps);
contains(spa, []) = false;
contains(spa, spa |> sps) = true;
spa != spb -> contains(spa, spb |> sps) = contains(spa, sps);
shuttles(fc) = if(fc == c1, [r1a,r2a,r3a], if(fc == c10, [r1b,r2b,r3b], []));
positions_sps([]) = [];
positions_sps(spa |> sps) = positions(spa) ++ positions_sps(sps);
connected_sps([]) = true;
connected_sps([spa]) = true;
connected_sps(spa |> spb |> sps) =

connected(spa, spb) && connected_sps(spb |> sps);
has_neighbour(sps, row_inc) = has_neighbour(rhead(sps), row_inc);
has_neighbour(sps, row_dec) = has_neighbour(head(sps), row_dec);
neighbour_sps(sps, row_inc) = neighbour(rhead(sps), row_inc);
neighbour_sps(sps, row_dec) = neighbour(head(sps), row_dec);
add_neighbour(sps, row_inc) = sps <| neighbour_sps(sps, row_inc);
add_neighbour(sps, row_dec) = neighbour_sps(sps, row_dec) |> sps;

% ShuttleOrientation operations
map

index: ShuttleOrientation -> Pos;
% index(soa) is the index of soa
lt: ShuttleOrientation # ShuttleOrientation -> Bool;
% lt(soa, sob) indicates if soa is less than sob
gt: ShuttleOrientation # ShuttleOrientation -> Bool;
% gt(soa, sob) indicates if soa is greater than sob
not: ShuttleOrientation -> ShuttleOrientation;
% not(soa) indicates the opposite of soa

var
soa, sob: ShuttleOrientation;

eqn
index(lowered) = 1;
index(tilted) = 2;
lt(soa, sob) = index(soa) < index(sob);

gt(soa, sob) = index(soa) > index(sob);
not(lowered) = tilted;
not(tilted) = lowered;

% Belt operations
map

index: R1Belt -> Pos;
index: R2Belt -> Pos;
index: R3Belt -> Pos;
% index(ba) is the index of belt ba
row: R1Belt -> FloorRow;
row: R2Belt -> FloorRow;
row: R3Belt -> FloorRow;
% row(ba) is the row of belt ba
lt: R1Belt # R1Belt -> Bool;
lt: R2Belt # R2Belt -> Bool;
lt: R3Belt # R3Belt -> Bool;
% lt(ba, bb) indicates if belt ba is less than belt bb
positions: R1Belt -> FloorPosList;
positions: R2Belt -> FloorPosList;
positions: R3Belt -> FloorPosList;
% positions(ba) represents the positions corresponding to belt ba
connected: R1Belt # R1Belt -> Bool;
connected: R2Belt # R2Belt -> Bool;
connected: R3Belt # R3Belt -> Bool;
% connected(ba, bb) indicates if the right of ba is connected to the left
% of bb
available: R1Belt # ShuttleState # LiftState -> Bool;
available: R2Belt # ShuttleState # LiftState -> Bool;
available: R3Belt # ShuttleState # LiftState -> Bool;
% available(ba, shs, ls) indicates if belt ba is available for states shs
% and ls

var
r1ba, r1bb: R1Belt;
r2ba, r2bb: R2Belt;
r3ba, r3bb: R3Belt;
fs: FloorState;
shs: ShuttleState;
ls: LiftState;
ms: MoveSize;

eqn
index(b_r1a_sh) = 1;
index(b_r1a) = 2;
index(b_r1lift) = 3;
index(b_r1b) = 4;
index(b_r1b_sh) = 5;
index(b_r2a_sh) = 1;
index(b_r2) = 2;
index(b_r2b_sh) = 3;
index(b_r3a_sh) = 1;
index(b_r3) = 2;
index(b_r3b_sh) = 3;
row(r1ba) = r1;
row(r2ba) = r2;
row(r3ba) = r3;
lt(r1ba, r1bb) = index(r1ba) < index(r1bb);
lt(r2ba, r2bb) = index(r2ba) < index(r2bb);
lt(r3ba, r3bb) = index(r3ba) < index(r3bb);
positions(b_r1a_sh) = [pos_r1(c1, pa), pos_r1(c1, pb)];
positions(b_r1a) = [pos_r1(c2, pa), pos_r1(c2, pb),

pos_r1(c3, pa), pos_r1(c3, pb),
pos_r1(c4, pa), pos_r1(c4, pb),
pos_r1(c5, pa), pos_r1(c5, pb),
pos_r1(c6, pa)];

positions(b_r1lift) = [pos_r1(c6, pb), pos_r1(c7, pa)];
positions(b_r1b) = [pos_r1(c7, pb),

pos_r1(c8, pa), pos_r1(c8, pb),
pos_r1(c9, pa), pos_r1(c9, pb)];

positions(b_r1b_sh) = [pos_r1(c10, pa), pos_r1(c10, pb)];
positions(b_r2a_sh) = [pos_r2(c1)];
positions(b_r2) = [pos_r2(c2),

pos_r2(c3),
pos_r2(c4),
pos_r2(c5),

pos_r2(c6),
pos_r2(c7),
pos_r2(c8),
pos_r2(c9)];

positions(b_r2b_sh) = [pos_r2(c10)];
positions(b_r3a_sh) = [pos_r3(c1)];
positions(b_r3) = [pos_r3(c2),

pos_r3(c3),
pos_r3(c4),
pos_r3(c5),
pos_r3(c6),
pos_r3(c7),
pos_r3(c8),
pos_r3(c9)];

positions(b_r3b_sh) = [pos_r3(c10)];
available(b_r1a_sh, shs, ls) = shs(r1a, lowered) == avail;
available(b_r1a, shs, ls) = true;
available(b_r1lift, shs, ls) = ls == ls_basement;
available(b_r1b, shs, ls) = true;
available(b_r1b_sh, shs, ls) = shs(r1b, lowered) == avail;
available(b_r2a_sh, shs, ls) = shs(r2a, lowered) == avail;
available(b_r2, shs, ls) = true;
available(b_r2b_sh, shs, ls) = shs(r2b, lowered) == avail;
available(b_r3a_sh, shs, ls) = shs(r3a, lowered) == avail;
available(b_r3, shs, ls) = true;
available(b_r3b_sh, shs, ls) = shs(r3b, lowered) == avail;

% BeltSet operations
map

is_set_b_r1: R1BeltSet -> Bool;
is_set_b_r2: R2BeltSet -> Bool;
is_set_b_r3: R3BeltSet -> Bool;
% is_set_b_rX(bs) indicates if bs is a set
% In the remainder of this data equations section each belt set bs has
% is_set_b_rX(bs) as a precondition.
contains: R1Belt # R1BeltSet -> Bool;
contains: R2Belt # R2BeltSet -> Bool;
contains: R3Belt # R3BeltSet -> Bool;
% contains(ba, bs) indicates if ba is an element of bs
positions_b_r1: R1BeltSet -> FloorPosList;
positions_b_r2: R2BeltSet -> FloorPosList;
positions_b_r3: R3BeltSet -> FloorPosList;
% positions_b_rX(bs) represents the floor positions corresponding to the
% belts in bs
connected_b_r1: R1BeltSet -> Bool;
connected_b_r2: R2BeltSet -> Bool;
connected_b_r3: R3BeltSet -> Bool;
% connected(bs) indicates if the belts in bs are connected
available_b_r1: R1BeltSet # ShuttleState # LiftState -> Bool;
available_b_r2: R2BeltSet # ShuttleState # LiftState -> Bool;
available_b_r3: R3BeltSet # ShuttleState # LiftState -> Bool;
% available_b_rX(bs, shs, ls) indicates if all belts in bs are available for
% states shs and ls

var
r1ba, r1bb: R1Belt;
r2ba, r2bb: R2Belt;
r3ba, r3bb: R3Belt;
r1bs: R1BeltSet;
r2bs: R2BeltSet;
r3bs: R3BeltSet;
fs: FloorState;
shs: ShuttleState;
ls: LiftState;

eqn
is_set_b_r1([]) = true;
is_set_b_r1(r1ba |> []) = true;
is_set_b_r1(r1ba |> r1bb |> r1bs) =

lt(r1ba, r1bb) && is_set_b_r1(r1bb |> r1bs);
is_set_b_r2([]) = true;
is_set_b_r2(r2ba |> []) = true;
is_set_b_r2(r2ba |> r2bb |> r2bs) =

lt(r2ba, r2bb) && is_set_b_r2(r2bb |> r2bs);
is_set_b_r3([]) = true;
is_set_b_r3(r3ba |> []) = true;

is_set_b_r3(r3ba |> r3bb |> r3bs) =
lt(r3ba, r3bb) && is_set_b_r3(r3bb |> r3bs);

contains(r1ba, []) = false;
contains(r1ba, r1ba |> r1bs) = true;
r1ba != r1bb -> contains(r1ba, r1bb |> r1bs) = contains(r1ba, r1bs);
contains(r2ba, []) = false;
contains(r2ba, r2ba |> r2bs) = true;
r2ba != r2bb -> contains(r2ba, r2bb |> r2bs) = contains(r2ba, r2bs);
contains(r3ba, []) = false;
contains(r3ba, r3ba |> r3bs) = true;
r3ba != r3bb -> contains(r3ba, r3bb |> r3bs) = contains(r3ba, r3bs);
positions_b_r1([]) = [];
positions_b_r1(r1ba |> r1bs) = positions(r1ba) ++ positions_b_r1(r1bs);
positions_b_r2([]) = [];
positions_b_r2(r2ba |> r2bs) = positions(r2ba) ++ positions_b_r2(r2bs);
positions_b_r3([]) = [];
positions_b_r3(r3ba |> r3bs) = positions(r3ba) ++ positions_b_r3(r3bs);
connected_b_r1([]) = true;
connected_b_r1(r1ba |> []) = true;
connected_b_r1(r1ba |> r1bb |> r1bs) =

connected(r1ba, r1bb) && connected_b_r1(r1bb |> r1bs);
connected_b_r2([]) = true;
connected_b_r2(r2ba |> []) = true;
connected_b_r2(r2ba |> r2bb |> r2bs) =

connected(r2ba, r2bb) && connected_b_r2(r2bb |> r2bs);
connected_b_r3([]) = true;
connected_b_r3(r3ba |> []) = true;
connected_b_r3(r3ba |> r3bb |> r3bs) =

connected(r3ba, r3bb) && connected_b_r3(r3bb |> r3bs);
connected(r1ba, r1bb) = succ(index(r1ba)) == index(r1bb);
connected(r2ba, r2bb) = succ(index(r2ba)) == index(r2bb);
connected(r3ba, r3bb) = succ(index(r3ba)) == index(r3bb);
available_b_r1([], shs, ls) = true;
available_b_r1(r1ba |> r1bs, shs, ls) =

available(r1ba, shs, ls) && available_b_r1(r1bs, shs, ls);
available_b_r2([], shs, ls) = true;
available_b_r2(r2ba |> r2bs, shs, ls) =

available(r2ba, shs, ls) && available_b_r2(r2bs, shs, ls);
available_b_r3([], shs, ls) = true;
available_b_r3(r3ba |> r3bs, shs, ls) =

available(r3ba, shs, ls) && available_b_r3(r3bs, shs, ls);

% Area operations
map

overlap: Area # Area -> Bool;
% area(a, b) indicates if area a and b overlap

var
fps, fpt: FloorPosList;
tc1a, tc1b, tc10a, tc10b: Bool;

eqn
overlap(area(fps, tc1a, tc1b), area(fpt, tc10a, tc10b)) =

overlap(fps, fpt) || (tc1a && tc1b) || (tc10a && tc10b);

% Instruction operations
map

index: Instruction -> Pos;
% index(i) is the index of instruction i
lt: Instruction # Instruction -> Bool;
% lt(i, j) indicates if i is less than j
valid: Instruction -> Bool;
% valid(i) indicates if instruction i is valid
area: Instruction -> Area;
% area(i) represents the area on which instruction i has any effect
overlap: Instruction # Instruction -> Bool;
% overlap(i, j) indicates if instruction i and j overlap

var
r1bs: R1BeltSet;
r2bs: R2BeltSet;
r3bs: R3BeltSet;
dc: DirCol;
ms: MoveSize;
sp: ShuttlePos;
sps: ShuttlePosSet;
so: ShuttleOrientation;

dr: DirRow;
lh: LiftHeight;
i,j: Instruction;

eqn
index(move_belts(r1bs, dc, ms)) = 1;
index(move_belts(r2bs, dc, ms)) = 2;
index(move_belts(r3bs, dc, ms)) = 3;
index(move_shuttles(sps, so, dr)) = 4;
index(tilt_shuttle(sp, so)) = 5;
index(move_lift(lh)) = 6;
index(rotate_lift) = 7;
lt(i, j) = index(i) < index(j);
valid(move_belts(r1bs, dc, ms)) = is_set_b_r1(r1bs);
valid(move_belts(r2bs, dc, ms)) = is_set_b_r2(r2bs);
valid(move_belts(r3bs, dc, ms)) = is_set_b_r3(r3bs);
valid(move_shuttles(sps, so, dr)) = is_set_sps(sps);
valid(tilt_shuttle(sp, so)) = true;
valid(move_lift(lh)) = true;
valid(rotate_lift) = true;
area(move_belts(r1bs, dc, ms)) = area(positions_b_r1(r1bs), false, false);
area(move_belts(r2bs, dc, ms)) = area(positions_b_r2(r2bs), false, false);
area(move_belts(r3bs, dc, ms)) = area(positions_b_r3(r3bs), false, false);
area(move_shuttles(sps, lowered, dr)) =

area(positions_sps(add_neighbour(sps, dr)), false, false);
area(move_shuttles(sps, tilted, dr)) =

area([], col(head(sps)) == c1, col(head(sps)) == c10);
area(tilt_shuttle(sp, so)) =

area(positions(sp), col(sp) == c1, col(sp) == c10);
area(move_lift(lh)) = area(positions(b_r1lift), false, false);
area(rotate_lift) = area(positions_b_r1([b_r1a, b_r1lift, b_r1b]), false, false);
overlap(i, j) = overlap(area(i), area(j));

% InstructionSet operations
map

is_set_is: InstructionSet -> Bool;
% is_set_is(is) indicates if is is a set
% In the remainder of the data equation section, for every InstructionSet is
% is_set_is(is) is a precondition.
valid: InstructionSet -> Bool;
% valid(is) indicates if the instructions in is are valid
overlap: Instruction # InstructionSet -> Bool;
% overlap(i, is) indicates if instruction i overlaps with any of the
% instructions in is

var
i,j: Instruction;
is: InstructionSet;

eqn
is_set_is([]) = true;
is_set_is(i |> []) = true;
is_set_is(i |> j |> is) = lt(i, j) && is_set_is(j |> is);
valid([]) = true;
valid(i |> is) = valid(i) && valid(is);
overlap(i, []) = false;
overlap(i, j |> is) = overlap(i, j) || overlap(i, is);

% AvailState operations
map

not: AvailState -> AvailState;
% not(asa) indicates the opposite of asa

eqn
not(avail) = n_avail;
not(n_avail) = avail;

% FloorState operations
map

init_fs: FloorState;
% init_fs is the initial floor state
update: FloorPos # OccState # FloorState -> FloorState;
cond_upd: FloorPos # OccState # FloorState -> FloorState;
ins_upd: FloorPos # OccState # FloorState -> FloorState;
% update(fpa, osa, fsa) represents fsa, where position fpa has value osa;
% cond_upd and ins_upd are auxiliary functions with the same meaning that
% are needed for efficiency
ins_upd_fps: FloorPosList # OccState # FloorState -> FloorState;

% ins_upd_fps(fps, osa, fsa) represents fsa, where positions fps have value
% osa
free: FloorPosList # FloorState -> Bool;
% free(fps, fsa) indicates that all positions in fps are free for state fsa
occupied: FloorPosList # FloorState -> Bool;
% occupied(fps, fsa) indicates that all positions in fps are occupied for
% state fsa
end_free: FloorPosList # DirCol # Bool # FloorState -> Bool;
% end_free(fps, dc, obl, fsa) indicates if the end of the list fps in
% direction dc is free for state fsa; if obl then the one but last element
% also needs to be free
even_occ: FloorPosList # FloorState -> Bool;
% even_occ(fps, fsa) indicates if the number of occupied positions in fps for
% state fsa is even
no_half_car: ShuttlePos # FloorState -> Bool;
% no_half_car(spa, fsa) indicates if there is a half car positioned on spa
% for state fsa
no_half_car_sps: ShuttlePosSet # FloorState -> Bool;
% no_half_car(sps, fsa) indicates if there is a half car positioned on any
% of the elements of sps for state fsa
shift_inc: FloorPosList # FloorState -> FloorState;
% shift_inc(fps, fsa) represents fsa, where the states of the positions in
% fps are shifted one position to the right; the first element in fps
% gets state free
gshift_inc: FloorPosList # FloorState # OccState -> FloorState;
% gshift_inc(fps, fsa, osa) has the same meaning as shift_inc(fps, fsa), with
% the exception that the first element in fps gets state osa
% (auxiliary function needed by shift_inc)
shift_dec: FloorPosList # FloorState -> FloorState;
% shift_dec(fps, fsa) represents fsa, where the states of the positions in
% fps are shifted one position to the left; the last element in fps gets
% state free
shift_inc_ctw: Bool # FloorPosList # FloorState -> FloorState;
% shift_inc_ctw(ctw, fps, fsa) represents fsa where the states of the
% positions in fps are shifted one position to the right;
% if ctw then two positions are shifted
shift_dec_ctw: Bool # FloorPosList # FloorState -> FloorState;
% shift_dec_ctw(ctw, fps, fsa) represents fsa where the states of the
% positions in fps are shifted one position to the left;
% if ctw then two positions are shifted
shift_inc_sps: ShuttlePosSet # FloorState -> FloorState;
% shift_inc(sps, fsa) represents fsa, where the states of the shuttles in
% sps are shifted one position downwards; the first element in sps
% gets state free
shift_dec_sps: ShuttlePosSet # FloorState -> FloorState;
% shift_dec(sps, fsa) represents fsa, where the states of the shuttles in
% sps are shifted one position upwards; the last element in sps gets state
% free

var
b: Bool;
fsa, fsb: FloorState;
fpa, fpb: FloorPos;
osa, osb: OccState;
fps: FloorPosList;
spa: ShuttlePos;
sps: ShuttlePosSet;

eqn
init_fs(fpa) = free;
fpa == fpb -> update(fpa, osa, fsa)(fpb) = osa;
fpa != fpb -> update(fpa, osa, fsa)(fpb) = fsa(fpb);
cond_upd(fpa, osa, fsa) =

if(fsa(fpa) == osa, fsa, update(fpa, osa, fsa));
ins_upd(fpa, osa, init_fs) = cond_upd(fpa, osa, init_fs);
lt(fpa,fpb) ->

ins_upd(fpa, osa, update(fpb, osb, fsa)) =
cond_upd(fpa, osa, update(fpb, osb, fsa));

gt(fpa,fpb) ->
ins_upd(fpa, osa, update(fpb, osb, fsa)) =
update(fpb, osb, ins_upd(fpa, osa, fsa));

fpa == fpb ->
ins_upd(fpa, osa, update(fpb, osb, fsa)) =
cond_upd(fpa, osa, fsa);

ins_upd_fps([], osa, fsa) = fsa;
ins_upd_fps(fpa |> fps, osa, fsa) =

ins_upd(fpa, osa, ins_upd_fps(fps, osa, fsa));
free([], fsa) = true;
free(fpa |> fps, fsa) = fsa(fpa) == free && free(fps, fsa);
occupied(fpa |> fps, fsa) = fsa(fpa) == occupied && occupied(fps, fsa);
occupied([], fsa) = true;
end_free(fps, col_inc, false, fsa) =

fsa(head(rev(fps))) == free;
end_free(fps, col_inc, true, fsa) =

fsa(head(l)) == free && fsa(head(tail(l))) == free
whr l = rev(fps) end;

end_free(fps, col_dec, false, fsa) =
fsa(head(fps)) == free;

end_free(fps, col_dec, true, fsa) =
fsa(head(fps)) == free && fsa(head(tail(fps))) == free;

even_occ([], fsa) = true;
fsa(fpa) == free ->

even_occ(fpa |> fps, fsa) = even_occ(fps, fsa);
fsa(fpa) == occupied ->

even_occ(fpa |> fps, fsa) = !even_occ(fps, fsa);
row(spa) != r1 ->

no_half_car(spa, fsa) = true;
row(spa) == r1 ->

no_half_car(spa, fsa) = even_occ(positions(spa), fsa);
no_half_car_sps([], fsa) = true;
no_half_car_sps(spa |> sps, fsa) =

no_half_car(spa, fsa) && no_half_car_sps(sps, fsa);
shift_inc(fps, fsa) = gshift_inc(fps, fsa, free);
gshift_inc([], fsa, osa) = fsa;
gshift_inc([fpa], fsa, osa) = ins_upd(fpa, osa, fsa);
gshift_inc(fpa |> fps, fsa, osa) =

ins_upd(fpa, osa, gshift_inc(fps, fsa, fsa(fpa)));
shift_dec([], fsa) = fsa;
shift_dec([fpa], fsa) = ins_upd(fpa, free, fsa);
shift_dec(fpa |> fpb |> fps, fsa) =

ins_upd(fpa, fsa(fpb), shift_dec(fpb |> fps, fsa));
shift_inc_ctw(false, fps, fsa) =

shift_inc(fps, fsa);
shift_inc_ctw(true, fps, fsa) =

shift_inc(fps, shift_inc(fps, fsa));
shift_dec_ctw(false, fps, fsa) =

shift_dec(fps, fsa);
shift_dec_ctw(true, fps, fsa) =

shift_dec(fps, shift_dec(fps, fsa));
shift_inc_sps([], fsa) = fsa;
row(spa) == r1 ->

shift_inc_sps(spa |> sps, fsa) =
ins_upd_fps(positions(spa), free,

shift_inc(positions_sps(spa |> sps), fsa));
row(spa) != r1 ->

shift_inc_sps(spa |> sps, fsa) = shift_inc(positions_sps(spa |> sps), fsa);
shift_dec_sps([], fsa) = fsa;
row(spa) == r1 ->

shift_dec_sps(spa |> sps, fsa) =
ins_upd_fps(positions(spa), fsa(head(positions(head(sps)))),

shift_dec(positions_sps(spa |> sps), fsa));
row(spa) != r1 ->

shift_dec_sps(spa |> sps, fsa) = shift_dec(positions_sps(spa |> sps), fsa);

% ShuttleState operations
map

init_shs: ShuttleState;
% initial shuttle state
update: ShuttlePos # ShuttleOrientation # AvailState # ShuttleState ->

ShuttleState;
cond_upd: ShuttlePos # ShuttleOrientation # AvailState # ShuttleState ->

ShuttleState;
ins_upd: ShuttlePos # ShuttleOrientation # AvailState # ShuttleState ->

ShuttleState;
% update(spa, soa, asa, fsa) represents fsa, where position fpa in
% orientation soa has value asa; cond_upd and ins_upd are auxiliary functions
% with the same meaning that are needed for efficiency
available: ShuttlePos # ShuttleOrientation # ShuttleState -> Bool;
% available(spa, soa, shs) indicates if shuttle spa is available in
% orientation soa for state shs

available_sps: ShuttlePosSet # ShuttleOrientation # ShuttleState -> Bool;
% available(sps, soa, shs) indicates if all shuttles in sps are available in
% orientation soa for state shs
shift_inc: ShuttlePosSet # ShuttleOrientation # ShuttleState -> ShuttleState;
% shift_inc(sps, soa, shs) represents shs, where the states of the shuttles
% in sps are shifted one shuttle downwards in orientation soa; the first
% element in sps gets state unavailable
gshift_inc: ShuttlePosSet # ShuttleOrientation # ShuttleState # AvailState

-> ShuttleState;
% gshift_inc(sps, soa, shs, asa) has the same meaning as shift_inc(sps, soa,
% shs), with the exception that the first element in sps gets state asa
% (auxiliary function needed by shift_inc)
shift_dec: ShuttlePosSet # ShuttleOrientation # ShuttleState -> ShuttleState;
% shift_dec(sps, soa, shs) represents shs, where the states of the shuttles
% in sps are shifted one shuttle upwards in orientation soa; the last
% element in sps gets state unavailable

var
b: Bool;
shs, shsa: ShuttleState;
spa, spb: ShuttlePos;
sps: ShuttlePosSet;
soa, sob: ShuttleOrientation;
asa, asb: AvailState;

eqn
init_shs(spa, lowered) = if(row(spa) != r3, avail, n_avail);
init_shs(spa, tilted) = if(row(spa) != r3, n_avail, avail);
spa == spb && soa == sob ->

update(spa, soa, asa, shs)(spb, sob) = asa;
spa != spb || soa != sob ->

update(spa, soa, asa, shs)(spb, sob) = shs(spb, sob);
cond_upd(spa, soa, asa, shs) =

if(shs(spa, soa) == asa, shs, update(spa, soa, asa, shs));
ins_upd(spa, soa, asa, init_shs) =

cond_upd(spa, soa, asa, init_shs);
lt(spa,spb) || (spa == spb && lt(soa, sob)) ->

ins_upd(spa, soa, asa, update(spb, sob, asb, shs)) =
cond_upd(spa, soa, asa, update(spb, sob, asb, shs));

gt(spa,spb) || (spa == spb && gt(soa, sob)) ->
ins_upd(spa, soa, asa, update(spb, sob, asb, shs)) =
update(spb, sob, asb, ins_upd(spa, soa, asa, shs));

spa == spb && soa == sob ->
ins_upd(spa, soa, asa, update(spb, sob, asb, shs)) =
cond_upd(spa, soa, asa, shs);

available(spa, soa, shs) = shs(spa, soa) == avail;
available_sps([], soa, shs) = true;
available_sps(spa |> sps, soa, shs) =

available(spa, soa, shs) && available_sps(sps, soa, shs);
shift_inc(sps, soa, shs) = gshift_inc(sps, soa, shs, n_avail);
gshift_inc([], soa, shs, asa) = shs;
gshift_inc([spa], soa, shs, asa) =

ins_upd(spa, soa, asa, shs);
gshift_inc(spa |> sps, soa, shs, asa) =

ins_upd(spa, soa, asa, gshift_inc(sps, soa, shs, shs(spa, soa)));
shift_dec([], soa, shs) = shs;
shift_dec([spa], soa, shs) = ins_upd(spa, soa, n_avail, shs);
shift_dec(spa |> spb |> sps, soa, shs) =

ins_upd(spa, soa, shs(spb, soa), shift_dec(spb |> sps, soa, shs));

% LiftState operations
map

height: LiftState -> LiftHeight;
% height(ls) represents the height of state ls
occupied: LiftState # FloorState -> Bool;
% occupied(ls, fs) indicates if the lift is occupieed for states ls and fs
make_ls: LiftHeight # OccState -> LiftState;
% make_ls(lh, os) represents the lift state corresponding to height lh and
% occupied state os

var
b: Bool;
fsa: FloorState;
osa: OccState;

eqn
height(lsf_street) = street;
height(lso_street) = street;

height(lsf_rotate) = rotate;
height(lso_rotate) = rotate;
height(ls_basement) = basement;
occupied(lsf_street, fsa) = false;
occupied(lso_street, fsa) = true;
occupied(lsf_rotate, fsa) = false;
occupied(lso_rotate, fsa) = true;
occupied(ls_basement, fsa) = occupied(positions(b_r1lift), fsa);
make_ls(street, occupied) = lso_street;
make_ls(street, free) = lsf_street;
make_ls(rotate, occupied) = lso_rotate;
make_ls(rotate, free) = lsf_rotate;
make_ls(basement, osa) = ls_basement;

% GlobalState operations
map

init_gs: GlobalState;
% init_gs is the initial global state
allowed: Instruction # GlobalState -> Bool;
% allowed(i, gs) indicates if instruction i is allowed given global state gs
nextstate: Instruction # ExecResult # GlobalState -> GlobalState;
% nextstate(i, r, gs) represents the global state after execution of
% instruction i with result r in state gs
allowed: InstructionSet # GlobalState -> Bool;
% allowed(is, gs) indicates if instruction set is is allowed given global
% state gs
nextstate: InstructionSet # ExecResult # GlobalState -> GlobalState;
% nextstate(is, r, gs) represents the global state after execution of
% instruction set is with result r in state gs
possible: Event # GlobalState -> Bool;
% possible(e, gs) indicates if event e is possible given global state gs
nextstate: Event # GlobalState -> GlobalState;
% nextstate(e, gs) represents the global state after the event e has occurred
% in state gs

var
gs, gsa: GlobalState;
fs, fsa: FloorState;
shs, shsa: ShuttleState;
ls, lsa: LiftState;
lh, lha: LiftHeight;
p,q: FloorPos;
s,t: OccState;
i,j: Instruction;
r1bs: R1BeltSet;
r2bs: R2BeltSet;
r3bs: R3BeltSet;
dc: DirCol;
ms: MoveSize;
sps: ShuttlePosSet;
dr: DirRow;
sp, spa, spb: ShuttlePos;
so: ShuttleOrientation;
is: InstructionSet;

eqn
init_gs =

glob_state(
init_fs,
init_shs,
lsf_street);

allowed(move_belts(r1bs, dc, ms), glob_state(fs, shs, ls)) =
r1bs != [] &&
connected_b_r1(r1bs) &&
available_b_r1(r1bs, shs, ls) &&
end_free(l, dc, ms == full, fs) &&
even_occ(l, fs)
whr l = positions_b_r1(r1bs) end;

allowed(move_belts(r2bs, dc, ms), glob_state(fs, shs, ls)) =
ms == full &&
r2bs != [] &&
connected_b_r2(r2bs) &&
available_b_r2(r2bs, shs, ls) &&
end_free(positions_b_r2(r2bs), dc, false, fs);

allowed(move_belts(r3bs, dc, ms), glob_state(fs, shs, ls)) =
ms == full &&

r3bs != [] &&
connected_b_r3(r3bs) &&
available_b_r3(r3bs, shs, ls) &&
end_free(positions_b_r3(r3bs), dc, false, fs);

allowed(move_shuttles(sps, so, dr), glob_state(fs, shs, ls)) =
sps != [] &&
connected_sps(sps) &&
available_sps(sps, so, shs) &&
% placing a comment in front of the following conjunct allows the shuttles
% to be moved even if a car is placed half on the shuttle
(so == lowered => no_half_car_sps(sps, fs)) &&
has_neighbour(sps, dr) &&
!available(neighbour_sps(sps, dr), so, shs);

allowed(tilt_shuttle(sp, lowered), glob_state(fs, shs, ls)) =
available_sps(shuttles(col(sp)), lowered, shs) &&
free(positions(sp), fs);

allowed(tilt_shuttle(sp, tilted), glob_state(fs, shs, ls)) =
shs(sp, tilted) == avail &&
shs(sp, lowered) == n_avail;

ls == ls_basement ->
allowed(move_lift(lh), glob_state(fs, shs, ls)) =
lh != height(ls) &&
% replacing the following conjunct by
% even_occ(positions(b_r1lift), fs);
% allows the lift to move up if two cars are placed half on the lift belt
if(free(positions(b_r1lift), fs), true,

even_occ(positions_b_r1([b_r1a_sh, b_r1a]), fs) &&
even_occ(positions_b_r1([b_r1b, b_r1b_sh]), fs));

ls != ls_basement ->
allowed(move_lift(lh), glob_state(fs, shs, ls)) =
lh != height(ls);

allowed(rotate_lift, glob_state(fs, shs, ls)) =
height(ls) == rotate &&
free([pos_r1(c5, pa), pos_r1(c5, pb), pos_r1(c6, pa),

pos_r1(c7, pb), pos_r1(c8, pa), pos_r1(c8, pb)], fs);
nextstate(move_belts(r1bs, col_inc, ms), ok, glob_state(fs, shs, ls)) =

glob_state(shift_inc_ctw(ms == full, positions_b_r1(r1bs), fs), shs, ls);
nextstate(move_belts(r1bs, col_dec, ms), ok, glob_state(fs, shs, ls)) =

glob_state(shift_dec_ctw(ms == full, positions_b_r1(r1bs), fs), shs, ls);
nextstate(move_belts(r2bs, col_inc, ms), ok, glob_state(fs, shs, ls)) =

glob_state(shift_inc(positions_b_r2(r2bs), fs), shs, ls);
nextstate(move_belts(r2bs, col_dec, ms), ok, glob_state(fs, shs, ls)) =

glob_state(shift_dec(positions_b_r2(r2bs), fs), shs, ls);
nextstate(move_belts(r3bs, col_inc, ms), ok, glob_state(fs, shs, ls)) =

glob_state(shift_inc(positions_b_r3(r3bs), fs), shs, ls);
nextstate(move_belts(r3bs, col_dec, ms), ok, glob_state(fs, shs, ls)) =

glob_state(shift_dec(positions_b_r3(r3bs), fs), shs, ls);
nextstate(move_shuttles(sps, lowered, row_inc), ok, glob_state(fs, shs, ls)) =

glob_state(shift_inc_sps(l, fs), shift_inc(l, lowered, shs), ls)
whr l = add_neighbour(sps, row_inc) end;

nextstate(move_shuttles(sps, lowered, row_dec), ok, glob_state(fs, shs, ls)) =
glob_state(shift_dec_sps(l, fs), shift_dec(l, lowered, shs), ls)

whr l = add_neighbour(sps, row_dec) end;
nextstate(move_shuttles(sps, tilted, row_inc), ok, glob_state(fs, shs, ls)) =

glob_state(fs, shift_inc(add_neighbour(sps, row_inc), tilted, shs), ls);
nextstate(move_shuttles(sps, tilted, row_dec), ok, glob_state(fs, shs, ls)) =

glob_state(fs, shift_dec(add_neighbour(sps, row_dec), tilted, shs), ls);
nextstate(tilt_shuttle(sp, so), ok, glob_state(fs, shs, ls)) =

glob_state(fs,
ins_upd(sp, so, n_avail, ins_upd(sp, not(so), avail, shs)),
ls);

occupied(ls, fs) ->
nextstate(move_lift(lh), ok, glob_state(fs, shs, ls)) =

glob_state(
if(height(ls) != basement && lh == basement,

ins_upd_fps(positions(b_r1lift), occupied, fs),
if(height(ls) == basement && lh != basement,

ins_upd_fps(positions(b_r1lift), free, fs),
fs
)

),
shs, make_ls(lh, occupied));

!occupied(ls, fs) ->
nextstate(move_lift(lh), ok, glob_state(fs, shs, ls)) =

glob_state(fs, shs, make_ls(lh, free));
nextstate(rotate_lift, ok, gs) = gs;
nextstate(i, fail, gs) = gs;
allowed([], gs) = false;
allowed(i |> [], gs) = allowed(i, gs);
allowed(i |> j |> is, gs) =

allowed(i, gs) &&
!overlap(i, j |> is) &&
allowed(j |> is, gs);

nextstate([], ok, gs) = gs;
nextstate(i |> is, ok, gs) = nextstate(i, ok, nextstate(is, ok, gs));
nextstate(is, fail, gs) = gs;
possible(add_car, glob_state(fs, shs, ls)) =

ls == lsf_street;
possible(remove_car, glob_state(fs, shs, ls)) =

ls == lso_street;
nextstate(add_car, glob_state(fs, shs, ls)) =

glob_state(fs, shs, lso_street);
nextstate(remove_car, glob_state(fs, shs, ls)) =

glob_state(fs, shs, lsf_street);

Appendix C: Processspeci®cations

In the following sections,the formal speci�cation of the processpart of the safetylayer is
given,followedby thespeci�cationsafterreducingcomplexity to enablesimulation.

C.1 General speci�cation

sort
ProcState = struct ps_idle | ps_await_state | ps_ack_deny

| ps_req | ps_exec | ps_ack_exec;
Layer = struct logical | safety | hardware;

act
snd_req, rcv_req, com_req: Layer # Layer # InstructionSet;
snd_ack_req, rcv_ack_req, com_ack_req: Layer # Layer # InstructionSet;
snd_deny_req, rcv_deny_req, com_deny_req: Layer # Layer # InstructionSet;
snd_ack_exec, rcv_ack_exec, ack_exec:Layer # Layer # InstructionSet # ExecResult;
snd_state, rcv_state, com_state: Layer # Layer # GlobalState;
snd_event, rcv_event, com_event: Layer # Layer # Event;

proc
SL(ps: ProcState, gs_sl: GlobalState, is: InstructionSet, r: ExecResult) =

(ps == ps_idle) ->
(sum isa: InstructionSet. valid(isa) ->

rcv_req(logical, safety, isa) .
SL(ps_ack_deny, gs_sl, isa, r)

)
+

(ps == ps_await_state) ->
(sum gs: GlobalState. rcv_state(hardware, safety, gs).

SL(ps_ack_deny, gs, is, r)
)

+
(ps == ps_ack_deny) ->

(allowed(is, gs_sl) ->
(snd_ack_req(safety, logical, is) . SL(ps_req, gs_sl, is, r)),
(snd_deny_req(safety, logical, is). SL(ps_idle, gs_sl, is, r)))

+
(ps == ps_req) ->

snd_req(safety, hardware, is) . SL(ps_exec, gs_sl, is, r)
+

(ps == ps_exec) ->
(sum ra: ExecResult.

rcv_ack_exec(safety, hardware, is, ra).
SL(ps_ack_exec, nextstate(is, ra, gs_sl), is, ra)

)
+

(ps == ps_ack_exec) ->
snd_ack_exec(safety, logical, is, r) . SL(ps_idle, gs_sl, is, r)

+
(ps == ps_idle) ->

(sum e: Event.
possible(e, gs_sl) -> rcv_event(hardware, safety, e) .

SL(ps, nextstate(e, gs_sl), is, r)
)

;

init
SL(ps_idle, init_gs, [], ok);

C.2 Speci�cation after reduction 1

In thefollowing speci�cation,instructionsetsarereplacedby singleinstructions.

sort
ProcState = struct ps_idle | ps_await_state | ps_ack_deny

| ps_req | ps_exec | ps_ack_exec;
Layer = struct logical | safety | hardware;

act
snd_req, rcv_req, com_req: Layer # Layer # Instruction;
snd_ack_req, rcv_ack_req, com_ack_req: Layer # Layer # Instruction;
snd_deny_req, rcv_deny_req, com_deny_req: Layer # Layer # Instruction;
snd_ack_exec, rcv_ack_exec, ack_exec:Layer # Layer # Instruction # ExecResult;
snd_state, rcv_state, com_state: Layer # Layer # GlobalState;
snd_event, rcv_event, com_event: Layer # Layer # Event;

proc
SL(ps: ProcState, gs_sl: GlobalState, i: Instruction, r: ExecResult) =

(ps == ps_idle) ->
(sum j: Instruction. valid(j) ->

rcv_req(logical, safety, j) .
SL(ps_ack_deny, gs_sl, j, r)

)
+

(ps == ps_await_state) ->
(sum gs: GlobalState. rcv_state(hardware, safety, gs).

SL(ps_ack_deny, gs, i, r)
)

+
(ps == ps_ack_deny) ->

(allowed(i, gs_sl) ->
(snd_ack_req(safety, logical, i) . SL(ps_req, gs_sl, i, r)),
(snd_deny_req(safety, logical, i). SL(ps_idle, gs_sl, i, r)))

+
(ps == ps_req) ->

snd_req(safety, hardware, i) . SL(ps_exec, gs_sl, i, r)
+

(ps == ps_exec) ->
(sum ra: ExecResult.

rcv_ack_exec(safety, hardware, i, ra).
SL(ps_ack_exec, nextstate(i, ra, gs_sl), i, ra)

)
+

(ps == ps_ack_exec) ->
snd_ack_exec(safety, logical, i, r) . SL(ps_idle, gs_sl, i, r)

+
(ps == ps_idle) ->

(sum e: Event.
possible(e, gs_sl) -> rcv_event(hardware, safety, e) .

SL(ps, nextstate(e, gs_sl), i, r)
)

;

init
SL(ps_idle, init_gs, move_lift(street), ok);

C.3 Speci�cation after reduction 2

Thefollowing speci�cationabstractsfrom non-essentialmessagepassing:

act
exec: Instruction;
occur: Event;

proc
S(gs_sl: GlobalState) =

sum i: Instruction. (valid(i) && allowed(i, gs_sl)) ->
exec(i) . S(nextstate(i, ok, gs_sl))

+
sum e: Event. possible(e, gs_sl) ->

occur(e) . S(nextstate(e, gs_sl))
;

init
S(init_gs);

Appendix D: Formal veri®cation

For the formal veri�cation of the requirementsthe speci�cation is reducedby changingthe
dataspeci�cationandaddingbehaviour to theprocessspeci�cation.

D.1 Speci�cation after reduction 3

The speci�cationafter reduction3 is the speci�cationwith the processspeci�cationof ap-
pendixC.3, togetherwith thedataspeci�cationof appendixB, in which thecapacitiesof the
beltsb r1a, b r1b, b r2 andb r3 arereduced.

positions(b_r1a) = [pos_r1(c2, pa), pos_r1(c2, pb), pos_r1(c6, pa)];
positions(b_r1b) = [pos_r1(c7, pb), pos_r1(c9, pa), pos_r1(c9, pb)];
positions(b_r2) = [pos_r2(c2), pos_r2(c9)];
positions(b_r3) = [pos_r3(c2), pos_r3(c9)];

D.2 Speci�cation after adding error actions

The requirementsof section6.1 are still ratherhigh-level. In order to formalise them in
mCRL2,we translatethemto amoredetailedlevel:

1. Belts:

(a) Foreachbeltb, beltsetbsandmovesizems, theinstructionmove belts(bs,col inc,
ms)shouldnotbeallowedif:

- Theright-mostpositionof belt b of sizemsis occupied.
- b is theright-mostbelt in bs.

Likewisefor theleft-mostpositionandbelt for instructionmove belts(bs,col dec,
ms).

(b) For eachshuttle s, belt set bs, direction d and move size ms, the instruction
move belts(bs,d, ms)shouldnotbeallowedif thefollowing conditionshold:

- Shuttles is availablein loweredpositionandcontains(a partof) a car.
- Thebelt correspondingto shuttles is in thebelt setbs.
- Directiond is pointingto thewall.
- Movesizemscouldmovethecarinto thewall.

(c) For eachbelt setbs, directiond andmove sizems, the instructionmove belts(bs,
d, ms)shouldnotbeallowedif bscontainsa belt thatis notavailable.

2. Shuttles:

(a) For all setsof shuttlepositionssps, orientationso anddirectionsd, theinstruction
move shuttles(sps,o, d) shouldnot beallowedif it containsa shuttlethatdirectly
facesawall in directiond.

(b) For all shuttlepositionss, setsof shuttlepositionsspsanddirectionsd, theinstruc-
tion move shuttles(sps,lowered,d) shouldnotbeallowedif:

- s is availablein loweredposition.
- s is theneighbourof spsin directiond.

(c) for all shuttless, setsof shuttlepositionsspsand directionsd, the instruction
move shuttles(sps,lowered,d) shouldnotbeallowedif:

- s is neithercompletelyfreenorcompletelyoccupied.
- s is anelementof thesetsps.

(d) For all shuttless andorientationso, the instructiontilt shuttle(s,o) shouldnot be
allowedif thereis a caron thebelt correspondingto s.

(e) For all shuttless andorientationso, the instructiontilt shuttle(s,o) shouldnot be
allowedif thereis a shuttlein positions in bothorientations.

3. Lift:

(a) For all heightsh theinstructionmove lift(h) shouldnotbeallowedif:

- Thelift is at thebasementlevel.
- Thelift containsoneor two carsthatareplacedhalf on thelift.

(b) Theinstructionrotate lift shouldnotbeallowedif oneof thefollowing holds:

- Thelift is notat rotateheight.
- The threehalf positionsbelts at the left and the right of the lift are not

completelyfree.

Theselow-level requirementsarespeci�ed in mCRL2asconditionsof erroractions.This is
expressedby thefollowing speci�cationthatis usedfor veri�cation.

sort
Requirement = struct req1a | req1b | req1c

| req2a | req2b | req2c | req2d | req2e
| req3a | req3b;

act
exec: Instruction;
occur: Event;
error: Requirement # Pos;

proc
V(gs_sl: GlobalState) =
% normal behaviour

sum i: Instruction. (valid(i) && allowed(i, gs_sl)) ->
exec(i) . V(nextstate(i, ok, gs_sl))

+
sum e: Event. possible(e, gs_sl) ->

occur(e) . V(nextstate(e, gs_sl))
% added behaviour with the purpose of checking the requirements;
% if this results in more behaviour, a requirement is not satisfied
% Requirement 1a
+

sum i: Instruction, b: R1Belt, ms: MoveSize, bs: R1BeltSet.
(!end_free(positions(b), col_inc, ms == full, fs(gs_sl)) &&

rhead(bs) == b &&
i == move_belts(bs, col_inc, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1a, 1) . delta
+

sum i: Instruction, b: R1Belt, ms: MoveSize, bs: R1BeltSet.
(!end_free(positions(b), col_dec, ms == full, fs(gs_sl)) &&

head(bs) == b &&
i == move_belts(bs, col_dec, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1a, 2) . delta
+

sum i: Instruction, b: R2Belt, ms: MoveSize, bs: R2BeltSet.
(!end_free(positions(b), col_inc, false, fs(gs_sl)) &&

rhead(bs) == b &&
i == move_belts(bs, col_inc, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1a, 3) . delta
+

sum i: Instruction, b: R2Belt, ms: MoveSize, bs: R2BeltSet.
(!end_free(positions(b), col_dec, false, fs(gs_sl)) &&

head(bs) == b &&
i == move_belts(bs, col_dec, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1a, 4) . delta
+

sum i: Instruction, b: R3Belt, ms: MoveSize, bs: R3BeltSet.
(!end_free(positions(b), col_inc, false, fs(gs_sl)) &&

rhead(bs) == b &&
i == move_belts(bs, col_inc, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1a, 5) . delta
+

sum i: Instruction, b: R3Belt, ms: MoveSize, bs: R3BeltSet.
(!end_free(positions(b), col_dec, false, fs(gs_sl)) &&

head(bs) == b &&
i == move_belts(bs, col_dec, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1a, 6) . delta
% Requirement 1b
+

sum i: Instruction, ms: MoveSize, bs: R1BeltSet.
(available(r1a, lowered, shs(gs_sl)) &&

!free(positions(r1a), fs(gs_sl)) &&
i == move_belts(bs, col_dec,

if(!occupied(positions(r1a), fs(gs_sl)), full, ms)) &&
contains(b_r1a_sh, bs) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1b, 1) . delta
+

sum i: Instruction, ms: MoveSize, bs: R1BeltSet.
(available(r1b, lowered, shs(gs_sl)) &&

!free(positions(r1b), fs(gs_sl)) &&
i == move_belts(bs, col_inc,

if(!occupied(positions(r1b), fs(gs_sl)), full, ms)) &&
contains(b_r1b_sh, bs) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1b, 2) . delta
+

sum i: Instruction, ms: MoveSize, bs: R2BeltSet.
(available(r2a, lowered, shs(gs_sl)) &&

!free(positions(r2a), fs(gs_sl)) &&
i == move_belts(bs, col_dec, ms) &&
contains(b_r2a_sh, bs) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1b, 3) . delta
+

sum i: Instruction, ms: MoveSize, bs: R2BeltSet.
(available(r2b, lowered, shs(gs_sl)) &&

!free(positions(r2b), fs(gs_sl)) &&
i == move_belts(bs, col_inc, ms) &&
contains(b_r2b_sh, bs) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1b, 4) . delta
+

sum i: Instruction, ms: MoveSize, bs: R3BeltSet.
(available(r3a, lowered, shs(gs_sl)) &&

!free(positions(r3a), fs(gs_sl)) &&
i == move_belts(bs, col_dec, ms) &&
contains(b_r3a_sh, bs) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1b, 5) . delta
+

sum i: Instruction, ms: MoveSize, bs: R3BeltSet.
(available(r3b, lowered, shs(gs_sl)) &&

!free(positions(r3b), fs(gs_sl)) &&
i == move_belts(bs, col_inc, ms) &&
contains(b_r3b_sh, bs) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1b, 6) . delta
% Requirement 1c
+

sum i: Instruction, dc: DirCol, ms: MoveSize, bs: R1BeltSet.
(!available_b_r1(bs, shs(gs_sl), ls(gs_sl)) &&

i == move_belts(bs, dc, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1c, 1) . delta
+

sum i: Instruction, dc: DirCol, ms: MoveSize, bs: R2BeltSet.
(!available_b_r2(bs, shs(gs_sl), ls(gs_sl)) &&

i == move_belts(bs, dc, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1c, 2) . delta
+

sum i: Instruction, dc: DirCol, ms: MoveSize, bs: R3BeltSet.
(!available_b_r3(bs, shs(gs_sl), ls(gs_sl)) &&

i == move_belts(bs, dc, ms) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req1c, 3) . delta
% Requirement 2a
+

sum i: Instruction, sps: ShuttlePosSet, so: ShuttleOrientation.
((contains(r1a, sps) || contains(r1b, sps)) &&

i == move_shuttles(sps, so, row_dec) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2a, 1) . delta
+

sum i: Instruction, sps: ShuttlePosSet, so: ShuttleOrientation.
((contains(r3a, sps) || contains(r3b, sps)) &&

i == move_shuttles(sps, so, row_inc) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2a, 2) . delta
% Requirement 2b
+

sum i: Instruction, sp: ShuttlePos, sps: ShuttlePosSet.
(available(sp, lowered, shs(gs_sl)) &&

connected(rhead(sps), sp) &&
i == move_shuttles(sps, lowered, row_inc) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2b, 1) . delta
+

sum i: Instruction, sp: ShuttlePos, sps: ShuttlePosSet.
(available(sp, lowered, shs(gs_sl)) &&

connected(sp, head(sps)) &&
i == move_shuttles(sps, lowered, row_dec) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2b, 2) . delta
% Requirement 2c
+

sum i: Instruction, sp: ShuttlePos, sps: ShuttlePosSet, dr: DirRow.
(!free(positions(sp),fs(gs_sl)) &&

!occupied(positions(sp),fs(gs_sl)) &&
contains(sp, sps) &&
i == move_shuttles(sps, lowered, dr) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2c, 1) . delta
% Requirement 2d
+

sum i: Instruction, sp: ShuttlePos, so: ShuttleOrientation.
(!free(positions(sp),fs(gs_sl)) &&

i == tilt_shuttle(sp, so) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2d, 1) . delta
% Requirement 2e
+

sum i: Instruction, sp: ShuttlePos, so: ShuttleOrientation.
(available(sp, so, shs(gs_sl)) &&

available(sp, not(so), shs(gs_sl)) &&
i == tilt_shuttle(sp, so) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req2e, 1) . delta
% Requirement 3a
+

sum i: Instruction, lh: LiftHeight.
(ls(gs_sl) == ls_basement &&

!free(positions(b_r1lift), fs(gs_sl)) &&
!(even_occ(positions_b_r1([b_r1a_sh, b_r1a]), fs(gs_sl)) &&

even_occ(positions_b_r1([b_r1b, b_r1b_sh]), fs(gs_sl))) &&
i == move_lift(lh) &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req3a, 1) . delta
% Requirement 3b
+

sum i: Instruction.
((height(ls(gs_sl)) != rotate ||

!free([pos_r1(c5, pa), pos_r1(c5, pb), pos_r1(c6, pa),
pos_r1(c7, pb), pos_r1(c8, pa), pos_r1(c8, pb)], fs(gs_sl))

) &&
i == rotate_lift &&
valid(i) &&
allowed(i, gs_sl)

) -> error(req3b, 1) . delta
;

init V(init_gs);

Generatedon13thOctober2005.

